NAG Library Routine Document

F08FFF (DORGTR)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F08FFF (DORGTR) generates the real orthogonal matrix Q, which was determined by F08FEF (DSYTRD) when reducing a symmetric matrix to tridiagonal form.

2 Specification

```
SUBROUTINE F08FFF (UPLO, N, A, LDA, TAU, WORK, LWORK, INFO)

INTEGER            N, LDA, LWORK, INFO
REAL (KIND=nag_wp) A(LDA,*), TAU(*), WORK(max(1,LWORK))
CHARACTER(1)       UPLO
```

The routine may be called by its LAPACK name ***dorgtr***.

3 Description

F08FFF (DORGTR) is intended to be used after a call to F08FEF (DSYTRD), which reduces a real symmetric matrix A to symmetric tridiagonal form T by an orthogonal similarity transformation: $A = QTQ^{\mathrm{T}}$. F08FEF (DSYTRD) represents the orthogonal matrix Q as a product of $n-1$ elementary reflectors.

This routine may be used to generate Q explicitly as a square matrix.

4 References

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

5 Parameters

1: UPLO – CHARACTER(1) *Input*

On entry: this **must** be the same parameter UPLO as supplied to F08FEF (DSYTRD).

Constraint: UPLO = 'U' or 'L'.

2: N – INTEGER *Input*

On entry: n, the order of the matrix Q.

Constraint: $\mathrm{N} \geq 0$.

3: A(LDA,$*$) – REAL (KIND=nag_wp) array *Input/Output*

Note: the second dimension of the array A must be at least $\max(1, \mathrm{N})$.

On entry: details of the vectors which define the elementary reflectors, as returned by F08FEF (DSYTRD).

On exit: the n by n orthogonal matrix Q.

4: LDA – INTEGER *Input*

On entry: the first dimension of the array A as declared in the (sub)program from which F08FFF (DORGTR) is called.

Constraint: LDA $\geq \max(1, \mathrm{N})$.

5: TAU$(*)$ – REAL (KIND=nag_wp) array *Input*

Note: the dimension of the array TAU must be at least $\max(1, \mathrm{N}-1)$.

On entry: further details of the elementary reflectors, as returned by F08FEF (DSYTRD).

6: WORK$(\max(1, \mathrm{LWORK}))$ – REAL (KIND=nag_wp) array *Workspace*

On exit: if INFO $= 0$, WORK(1) contains the minimum value of LWORK required for optimal performance.

7: LWORK – INTEGER *Input*

On entry: the dimension of the array WORK as declared in the (sub)program from which F08FFF (DORGTR) is called.

If LWORK $= -1$, a workspace query is assumed; the routine only calculates the optimal size of the WORK array, returns this value as the first entry of the WORK array, and no error message related to LWORK is issued.

Suggested value: for optimal performance, LWORK $\geq (\mathrm{N}-1) \times nb$, where nb is the optimal ***block size***.

Constraint: LWORK $\geq \max(1, \mathrm{N}-1)$ or LWORK $= -1$.

8: INFO – INTEGER *Output*

On exit: INFO $= 0$ unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If INFO $= -i$, argument i had an illegal value. An explanatory message is output, and execution of the program is terminated.

7 Accuracy

The computed matrix Q differs from an exactly orthogonal matrix by a matrix E such that

$$\|E\|_2 = O(\epsilon),$$

where ϵ is the ***machine precision***.

8 Further Comments

The total number of floating point operations is approximately $\frac{4}{3}n^3$.

The complex analogue of this routine is F08FTF (ZUNGTR).

9 Example

This example computes all the eigenvalues and eigenvectors of the matrix A, where

$$A = \begin{pmatrix} 2.07 & 3.87 & 4.20 & -1.15 \\ 3.87 & -0.21 & 1.87 & 0.63 \\ 4.20 & 1.87 & 1.15 & 2.06 \\ -1.15 & 0.63 & 2.06 & -1.81 \end{pmatrix}.$$

Here A is symmetric and must first be reduced to tridiagonal form by F08FEF (DSYTRD). The program then calls F08FFF (DORGTR) to form Q, and passes this matrix to F08JEF (DSTEQR) which computes the eigenvalues and eigenvectors of A.

9.1 Program Text

```
    Program f08fffe

!     F08FFF Example Program Text

!     Mark 24 Release. NAG Copyright 2012.

!     .. Use Statements ..
      Use nag_library, Only: dorgtr, dsteqr, dsytrd, f06qff, nag_wp, x04caf
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter               :: nin = 5, nout = 6
!     .. Local Scalars ..
      Integer                          :: i, ifail, info, lda, ldz, lwork, n
      Character (1)                    :: uplo
!     .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable  :: a(:,:), d(:), e(:), tau(:), work(:), &
                                          z(:,:)
!     .. Executable Statements ..
      Write (nout,*) 'F08FFF Example Program Results'
!     Skip heading in data file
      Read (nin,*)
      Read (nin,*) n
      lda = n
      ldz = n
      lwork = 64*n
      Allocate (a(lda,n),d(n),e(n),tau(n),work(lwork),z(ldz,n))

!     Read A from data file

      Read (nin,*) uplo
      If (uplo=='U') Then
        Read (nin,*)(a(i,i:n),i=1,n)
      Else If (uplo=='L') Then
        Read (nin,*)(a(i,1:i),i=1,n)
      End If

!     Reduce A to tridiagonal form T = (Q**T)*A*Q
!     The NAG name equivalent of dsytrd is f08fef
      Call dsytrd(uplo,n,a,lda,d,e,tau,work,lwork,info)

!     Copy A into Z
      Call f06qff(uplo,n,n,a,lda,z,ldz)

!     Form Q explicitly, storing the result in Z
!     The NAG name equivalent of dorgtr is f08fff
      Call dorgtr(uplo,n,z,ldz,tau,work,lwork,info)

!     Calculate all the eigenvalues and eigenvectors of A
!     The NAG name equivalent of dsteqr is f08jef
      Call dsteqr('V',n,d,e,z,ldz,work,info)

      Write (nout,*)
      If (info>0) Then
```

```
        Write (nout,*) 'Failure to converge.'
      Else

!       Print eigenvalues and eigenvectors

        Write (nout,*) 'Eigenvalues'
        Write (nout,99999) d(1:n)
        Write (nout,*)
        Flush (nout)

!       ifail: behaviour on error exit
!              =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
        ifail = 0
        Call x04caf('General',' ',n,n,z,ldz,'Eigenvectors',ifail)

      End If

99999 Format (3X,(8F8.4))
    End Program f08fffe
```

9.2 Program Data

```
F08FFF Example Program Data
  4                           :Value of N
  'L'                         :Value of UPLO
  2.07
  3.87  -0.21
  4.20   1.87   1.15
 -1.15   0.63   2.06  -1.81   :End of matrix A
```

9.3 Program Results

```
 F08FFF Example Program Results

 Eigenvalues
    -5.0034 -1.9987  0.2013  8.0008

 Eigenvectors
          1       2       3       4
 1   0.5658 -0.2328 -0.3965  0.6845
 2  -0.3478  0.7994 -0.1780  0.4564
 3  -0.4740 -0.4087  0.5381  0.5645
 4   0.5781  0.3737  0.7221  0.0676
```